Generic Downwards Accumulations

Jeremy Gibbons

School of Computing and Mathematical Sciences, Oxford Brookes University, Gipsy Lane, Headington, Oxford OX3 0BP, UK.

Email jgibbons@brookes.ac.uk.

Abstract. A downwards accumulation is a higher-order operation that distributes information downwards through a data structure, from the root towards the leaves. The concept was originally introduced in an ad hoc way for just a couple of kinds of tree. We generalize the concept to an arbitrary regular datatype; the resulting definition is co-inductive.

1 Introduction

The notion of scans or accumulations on lists is well known, and has proved very fruitful for expressing and calculating with programs involving lists [4]. Gibbons [7, 8] generalizes the notion of accumulation to various kinds of tree; that generalization too has proved fruitful, underlying the derivations of a number of tree algorithms, such as the parallel prefix algorithm for prefix sums [15, 8], Reingold and Tilford’s algorithm for drawing trees tidily [21, 9], and algorithms for query evaluation in structured text [16, 23].

There are two varieties of accumulation on lists: leftwards and rightwards. Leftwards accumulation labels every node of the list with some function of its successors—the tail segment starting at that node—thereby passing information from right to left along the list; rightwards accumulation labels every node with some function of its predecessors—the initial segment ending at that node— passing information from left to right. Similarly, there are two varieties of accumulation on trees: upwards and downwards. Upwards accumulation labels every node with some function of its descendants—the subtree rooted at that node— thereby passing information up the tree; downwards accumulation labels every node with some function of its ancestors—the path from the root to that node— passing information down the tree.

A flaw in the definitions of accumulations on trees from [7, 8] is that they are rather ad hoc. There is no formal relationship between accumulations on different kinds of tree, so each new kind of tree has to be considered from scratch. A recent trend in constructive algorithmics has been the development of theories

Copyright © Jeremy Gibbons, 1999. This paper is a revised version of Polytypic Downwards Accumulations, in LNCS 1422: Mathematics of Program Construction, ed. Johan Jeuring, 1998.

of generic [12, 13] or polytypic [14] operations, parameterized by a datatype. Another name for this kind of abstraction is higher-order polymorphism. A generic program in this sense eliminates the unwanted ad-hockery. The categorical approach to datatypes popularized by Malcolm [17] is an early example of generic programming: it allows a single unified definition of operations such as map, fold and unfold, parameterized by the datatype concerned.

Bird et al. [1] generalize upwards accumulation to an arbitrary regular datatype, unifying the previous ad hoc definitions. In this paper, we generalize downwards accumulation to an arbitrary regular datatype too. This is a more difficult problem: whereas the descendants of a node in a data structure (some kind of tree) form another data structure of the same type (another tree), the ancestors of a node are in general of a completely different type (a ‘path’).

We conclude this introductory section with a summary of notation. The remainder of the paper is structured as follows. In Section 2 we recall the monotypic definitions of accumulations on lists and trees. We briefly summarize the theory of datatypes in Section 3. In Section 4 we discuss Bird et al.’s generic definition of upwards accumulation. In Section 5 we develop a generic definition of downwards accumulation, and in Section 6 we collect some properties of generic downwards accumulations. Section 7 concludes.

1.1 Functions

The type judgement ‘a :: A’ declares that value a is of type A; the type A → B denotes the type of functions from A to B. Function application is denoted by juxtaposition; the identity function is written id, so that id a = a for every a. The unit type 1 has just one element, denoted (); there is a unique total function one :: A →1 for every type A. The function const :: A →B →A ignores its second argument and always returns its first, that is, it satisfies const a b = a for all a and b.

1.2 The Pair Calculus

The functors + and × denote separated sum and cartesian product respectively; × binds tighter than +. The product projections are fst and snd, and the product morphism (sometimes called ‘split’ or ‘fork’) f △ g has type A → B × C when f :: A → B and g :: A → C .

1.3 Lists

The elements of the type List(A) are finite sequences of elements of type A. The empty list is denoted [ ], singleton lists [a], and ++ is list concatenation.

2 Accumulations

To provide motivation and intuition for what follows, we review here the ‘monotypic’ definitions of leftwards and rightwards accumulations on lists (well-known from functional programming) and upwards and downwards accumulations on binary trees (as presented in [7, 8]). Bird et al.’s generalization of upwards accumulation and our generalization of downwards accumulation, when specialized to the same types, reduce essentially to these monotypic definitions.

2.1 Accumulations on Lists

The standard leftwards and rightwards accumulations on lists are defined as follows. Leftwards accumulation distributes information from right to left along a list. It is traditionally, if confusingly, called scanr, perhaps because the parentheses collect at the right-hand end of the list.

scanr :: (a->b->b) -> b -> List a -> List b
scanr f e [] = [e]
scanr f e (a:x) = f a (head x’) : x’ where x’ = scanr f e x

(Here, ‘a:x’ is the non-empty list with head a and tail x, and the function head satisfies head (a:x) = a. The notation is essentially that of Haskell [20].) For example,

scanr (+) 0 [1,2,3] = [6,5,3,0]

(Here, ‘(+)’ is the addition operator passed as an argument, and ‘[1,2,3]’ a list with three elements.)

Rightwards accumulation, of course, distributes information in the opposite direction:

scanl :: (b->a->b) -> b -> List a -> List b
scanl f e [] = [e]
scanl f e (a:x) = e : scanl f (f e a) x

For example,

scanl (+) 0 [1,2,3] = [0,1,3,6]

To make the analogy with trees clearer, we will first adapt the standard list accumulations to operate on a type of non-empty lists:

data Plist a = Wrap a | ConsP a (Plist a)

Informally, a non-empty list is either a singleton, Wrap a, or an element prefixed onto another non-empty list, ConsP a x. On this type, the accumulations return a list of the same length as the argument, instead of a list one element longer. As will become clear, leftwards accumulation on lists is a kind of upwards accumulation, so we rename the function to uaPlist:

uaPlist :: (a->b) -> (a->b->b) -> Plist a -> Plist b
uaPlist f g (Wrap a) = Wrap (f a)
uaPlist f g (ConsP a x) = ConsP (g a (headPlist x’)) x’
                          where x’ = uaPlist f g x where headPlist returns the first element of a non-empty list:

headPlist :: Plist a -> a
headPlist (Wrap a) = a
headPlist (ConsP a x) = a

For example,

uaPlist id (+) [1,2,3] = [6,5,3]

Rightwards accumulation (which we now call daPlist, for ‘downwards accumulation on a list’) is generalized with an extra parameter, so that it is not the accumulating parameter e itself that appears in the result, but some function

f e a of e and the label a at this node. The accumulating parameter for the

next node is g e a, another function of the accumulating parameter and the label at this node.

daPlist :: (c->a->b) -> (c->a->c) -> c -> Plist a -> Plist b
daPlist f g e (Wrap a) = Wrap (f e a)
daPlist f g e (ConsP a x) = ConsP (f e a) (daPlist f g (g e a) x)

For example,

daPlist (+) (+) 0 [1,2,3] = [1,3,6]

A more complicated example, making use of the generalization allowing the accumulating parameter to be of a different type than the result labels, is to compute the ‘running averages’ of a list:

averages = daPlist f g (0,0) where f (s,n) a = (s+a)/(n+1)
                                   g (s,n) a = (s+a , n+1)

2.2 Accumulations on Binary Trees

The elements of the datatype of homogeneous binary trees are trees with internal and external labels of the same type:

data Tree a = Leaf a | Bin a (Tree a) (Tree a)

Thus, a tree is either a Leaf with a label, or a Bin with a label and two subtrees.

Upwards accumulation on these trees is fairly straightforward:

uaTree :: (a->b) -> (a->b->b->b) -> Tree a -> Tree b
uaTree f g (Leaf a) = Leaf (f a)
uaTree f g (Bin a t u) = Bin (g a (root t’) (root u’)) t’ u’
                         where t’ = uaTree f g t
                               u’ = uaTree f g u

where root returns the root label of a tree:

root :: Tree a -> a
root (Leaf a) = a
root (Bin a t u) = a

The archetypical example of an upwards accumulation is to count the descendants of every node:

sizes :: Tree a -> Tree Int
sizes = uaTree one plus where one a = 1
                               plus a m n = 1+m+n

Downwards accumulation is a little trickier: we need two functions for generating new accumulating parameters, so that left children can be treated differently from right children.

daTree :: (c->a->b) -> (c->a->c) -> (c->a->c) -> c -> Tree a -> Tree b
daTree f g h e (Leaf a) = Leaf (f e a)
daTree f g h e (Bin a t u) = Bin (f e a) (daTree f g h (g e a) t)
                                         (daTree f g h (h e a) u)

The archetypical example of a downwards accumulation is to count the ancestors of every node:

depths :: Tree a -> Tree Int
depths = daTree label next next 0 where label e a = e+1
                                         next e a = e+1

3 Datatypes

In this section, we briefly review the construction of regular datatypes, in the style of Malcolm [17]. However, we assume a setting of continuous functions between pointed complete partial orders, in contrast to Malcolm’s setting of total functions between sets. Given a bifunctor F , the datatype T is the type functor such that the type T (A) is isomorphic to F (A, T (A)); the isomorphism is provided by the constructor in_F :: F (A, T (A)) → T (A) and the destructor out_F :: T (A) → F (A, T (A)), both of which are strict. (Fokkinga and Meijer [6], building on the work of Reynolds [22], show that this determines T up to isomorphism in terms of F .) We call T (A) the canonical fixpoint of the functor F (A, ).

A strict function ϕ :: F (A, B) → B induces a fold (or ‘catamorphism’) fold_F ϕ :: T (A) → B, with the universal property

h = fold_F ϕ ⇔ h ∘ in_F = ϕ ∘ F (id, h)

A (not necessarily strict) function ψ :: B → F (A, B) induces an unfold (or ‘anamorphism’) unfold_F ψ :: B → T (A), with the universal property

h = unfold_G ψ ⇔ out_G ∘ h = G(id, h) ∘ ψ

The function map_F f :: T (A) → T (B) for f :: A → B applies f to every element of its argument; in other words, map_F f is the action T (f) of the functor T on f. It can be defined either as a fold or as an unfold:

map_F f = fold_F (in_F ∘ F (f, id))

= unfold_F (F (f, id) ∘ out_F )

Example 1. We will use the following datatypes as running examples throughout the paper.

1. Cons lists are constructed from the functor F (A, X ) = 1 + A × X :

data List a = Nil | Cons a (List a)

The corresponding fold is

foldL :: (Either () (a,b) -> b) -> List a -> b
foldL phi Nil = phi (Left ())
foldL phi (Cons a x) = phi (Right (a, foldL phi x))

(Here, the Haskell datatype Either is defined by

data Either a b = Left a | Right b

and corresponds roughly to disjoint sum.) For example, the function sizeL, which returns the size (length) of a list, is a fold:

sizeL :: List a -> Int
sizeL = foldL phiSizeL
phiSizeL (Left ()) = 0
phiSizeL (Right (a,n)) = 1+n

(We will use phiSizeL later.)

2. Leaf-labelled binary trees are built from the functor F (A, X ) = A + X × X :

data Ltree a = LeafT a | BinT (Ltree a) (Ltree a)

The corresponding fold is

foldT :: (Either a (b,b) -> b) -> Ltree a -> b
foldT phi (LeafT a) = phi (Left a)
foldT phi (BinT t u) = phi (Right (foldT phi t, foldT phi u))

The function sizeT, which returns the size of (that is, the number of elements in) a leaf-labelled binary tree, is a fold:

sizeT :: Ltree a -> Int
sizeT = foldT phiSizeT
phiSizeT (Left a) = 1
phiSizeT (Right (m,n)) = m+n

3. Internally-labelled binary trees are constructed from the functor F (A, X ) = 1 + A × X × X :

data Btree a = Empty | BinB a (Btree a) (Btree a)

The corresponding fold is

foldB :: (Either () (a,b,b) -> b) -> Btree a -> b
foldB phi Empty = phi (Left ())
foldB phi (BinB a t u) =
  phi (Right (a, foldB phi t, foldB phi u))

The function sizeB is defined as follows:

sizeB :: Btree a -> Int
sizeB = foldB phiSizeB
phiSizeB (Left ()) = 0
phiSizeB (Right (a,m,n)) = 1+m+n

4. Rose trees [18] are constructed from the functor F (A, X ) = A × List(X ):

data Rtree a = ConsR a (List (Rtree a))

A tree of type Rtree a consists of a label of type a and a list of children. The corresponding fold is

foldR :: ((a, List b) -> b) -> Rtree a -> b
foldR phi (ConsR a ts) = phi (a, map (foldR phi) ts)

where map f xs applies function f to every element of list xs. The function

sizeR is defined as follows:
  sizeR :: Rtree a -> Int
  sizeR = foldR phiSizeR
  phiSizeR :: (a, List Int) -> Int
  phiSizeR (a,ns) = 1 + sum ns

where sum sums a list of integers. ⊓⊔ 4 Generic Upwards Accumulations

Bird et al. [1] generalize upwards accumulation to an arbitrary regular datatype. We summarize their construction here. It is related to, but not the same as, Meertens’ generic definition of the ‘predecessors’ of a data structure [19]. This section serves partly as motivation and to provide intuition for the definitions in Section 5, as upwards accumulations are simpler than downwards accumulations. More importantly, however, we will use part of their construction in our own. Throughout this section, we assume that F is a bifunctor, and that T (A) is the canonical fixpoint of F (A, ).

4.1 Labelled Types

The essential idea is that an upwards accumulation of a data structure x computes all the partial results involved in folding x. Each partial result is the fold of some subtree of x, and is stored in the resulting data structure at the root of that subtree.

A consequence of taking this approach is that the result type of an upwards accumulation may be different from the argument type: the result type has a label at every node, whereas the argument type need not. For example, an upwards accumulation on a leaf-labelled binary tree should produce a homogeneous binary tree. Bird et al. call the datatype of homogeneous binary trees the labelled variant of the type of leaf-labelled binary trees, and give the following general construction for it.

Definition 2. The labelled type L(A) corresponding to the datatype T (A) is the canonical fixpoint of the functor G(A, ), where G is defined by G(A, X ) = A × F (1, X ). ⊓⊔

Informally, F (1, X ) is like F (A, X ), but with all the labels (of type A) removed; thus, using A × F (1, X ) ensures that every node carries precisely one label.

Example 3.

1. The datatype of cons lists is constructed from the functor F (A, X ) = 1 + A × X , so the functor G is given by

G(A, X ) = A × F (1, X )

= A × (1 + 1 × X ) ≈ A × (1 + X )

This induces a labelled type of non-empty lists:

data Nlist a = ConsN a (Either () (Nlist a))

This type is similar to the datatype Plist from Section 2. The unfold for this type is

unfoldN :: (b -> (a, Either () b)) -> b -> Nlist a
unfoldN phi b = case phi b of
  (a, Left ()) -> ConsN a (Left ())
  (a, Right b’) -> ConsN a (Right (unfoldN phi b’))

(We give folds for the ‘ordinary’ datatypes and unfolds for the labelled variants, because an accumulation consumes an ordinary datatype and produces a labelled one.)

2. The datatype of leaf-labelled binary trees is constructed from the functor F (A, X ) = A + X × X , so the functor G is given by

G(A, X ) = A × F (1, X )

= A × (1 + X × X )

This induces the labelled type of homogeneous binary trees, as desired:

data Htree a = ConsH a (Either () (Htree a,Htree a))

(This type is similar to the datatype Tree of Section 2.) The corresponding unfold is

unfoldH :: (b -> (a, Either () (b,b))) -> b -> Htree a
unfoldH phi b = case phi b of
  (a, Left ()) -> ConsH a (Left ())
  (a, Right (b’,b’’)) -> ConsH a (Right (unfoldH phi b’,
                                         unfoldH phi b’’))

3. The datatype of internally-labelled binary trees is constructed from the functor F (A, X ) = 1 + A × X × X , so the functor G is given by

G(A, X ) = A × F (1, X )

= A × (1 + 1 × X × X ) ≈ A × (1 + X × X )

and induces the labelled type of homogeneous binary trees, just as for leaflabelled binary trees.

4. The datatype of homogeneous binary trees is constructed from the functor F (A, X ) = A × (1 + X × X ), so the functor G is given by

G(A, X ) = A × F (1, X )

= A × (1 × (1 + X × X )) ≈ A × (1 + X × X ) = F (A, X )

and so homogeneous binary trees are their own labelled type. We call a type that is (isomorphic to) its own labelled variant homogeneous, and a non-homogeneous type heterogeneous. In general, if F (A, X ) = A × F^′(X ) for some F^′ independent of A, then G(A, X ) = A × F (1, X ) = A × 1 × F^′(X ) ≈ F (A, X ), and so the generated type is homogeneous. In particular, ‘constructing the labelled variant’ is an idempotent operation.

5. The datatype of rose trees is constructed from the functor F (A, X ) = A × List(X ), so the functor G is given by

G(A, X ) = A × F (1, X )

= A × (1 × List(X )) ≈ A × List(X ) = F (A, X )

and so rose trees are homogeneous. The corresponding unfold is

unfoldR :: (b -> (a, List b)) -> b -> Rtree a
unfoldR phi b = case phi b of
  (a,bs) -> ConsR a (map (unfoldR phi) bs)

(The purpose of these and subsequent examples is to present definitions that the rational programmer might have come up with after some thought, and to show that those definitions are instantiations of our constructions. We use the term ‘instantiation’ a little loosely: literal instantiations often involve extra superfluous units, which we have suppressed. For example, the literal use of the characterization G(A, X ) = A × F (1, X ) gives for cons lists the instantiation G(A, X ) = A × (1 + 1 × X ) and the corresponding type declaration

data Nlist a = ConsN a (Either () ((),Nlist a))

but in Example 3.1 we suppressed the superfluous ().) ⊓⊔

4.2 Upwards Accumulations

The upwards accumulation ua_F ϕ constructs a labelled data structure of type L(B) from a data structure of type T (A); the argument ϕ is a suitable argument for fold_F , and the data structure returned by ua_F ϕ x contains all the partial results obtained in the process of computing fold_F ϕ x.

We will define ua_F ϕ as an unfold, namely ua_F ϕ = unfold_G ψ for some ψ dependent on ϕ. Type considerations reveal that

ψ :: T (A) → G(B, T (A))

:: T (A) → B × F (1, T (A))

Such a ψ is necessarily of the form ψ_1 △ ψ_2 for some ψ_1 :: T (A) → B and ψ_2 :: T (A) → F (1, T (A)). We let ψ_1 be simply the corresponding fold:

ψ_1 = fold_F ϕ

and define ψ_2 to destruct the node and discard the root labels:

ψ_2 = F (one, id) ∘ out_F

Thus, we have the following definition.

Definition 4. Upwards accumulation ua_F :: (F (A, B) → B) → T (A) → L(B) is defined by

ua_F ϕ = unfold_G (fold_F ϕ △ (F (one, id) ∘ out_F )) ⊓⊔ This definition is illustrated in Figure 1. The dotted arrow is the product morphism induced by the two independent parts of the diagram.

T (A) outF F (A, T (A))

B fst fold^F ϕB × F (1, T (A)) = G(B, T (A)) foldF ϕ △ F (one, id) ∘ outF

. . . . . . snd F (1, T (A))

F (one, id)

Fig. 1. The anatomy of an upwards accumulation

Example 5.

1. The labelled variant of cons lists is non-empty cons lists, so uaL is defined in terms of unfoldN:

uaL :: (Either () (a,b) -> b) -> List a -> Nlist b
uaL phi = unfoldN psi
  where psi Nil = (foldL phi Nil, Left ())
        psi (Cons a x) = (foldL phi (Cons a x), Right x)

For example, the function sizesL labels every node of a list with the length of the tail segment starting at that node (returning a list of the form [n,...,0]):

sizesL :: List a -> Nlist Int
sizesL = uaL phiSizeL

where phiSizeL is as defined in Example 1.1.

2. The labelled variant of leaf-labelled binary trees is homogeneous binary trees, so we use unfoldH:

uaT :: (Either a (b,b) -> b) -> Ltree a -> Htree b
uaT phi = unfoldH psi
  where psi (LeafT a) = (foldT phi (LeafT a), Left ())
        psi (BinT t u) = (foldT phi (BinT t u), Right (t, u))

For example, the function sizesT labels every node of a leaf-labelled binary tree with the size of the subtree rooted there:

sizesT :: Ltree a -> Htree Int
sizesT = uaT phiSizeT

3. For internally-labelled binary trees too, the labelled variant is homogeneous binary trees:

uaB :: (Either () (a,b,b) -> b) -> Btree a -> Htree b
uaB phi = unfoldH psi
  where psi Empty = (foldB phi Empty, Left ())
        psi (BinB a t u) = (foldB phi (BinB a t u), Right (t, u))

The corresponding ‘sizes’ function on these trees is

sizesB :: Btree a -> Htree Int
sizesB = uaB phiSizeB

4. Rose trees are homogeneous, so we use unfoldR:

uaR :: ((a, List b) -> b) -> Rtree a -> Rtree b
uaR phi = unfoldR psi
  where psi (ConsR a ts) = (foldR phi (ConsR a ts), ts)

The corresponding ‘sizes’ function is

sizesR :: Rtree a -> Rtree Int
sizesR = uaR phiSizeR

⊓⊔

Of course, taken literally, Definition 4 makes rather an inefficient program: the substructures rooted at each node are folded independently, without exploiting the fact that the results of folding the children of a node can be reused in folding the substructure rooted at the node itself. Fortunately, an immediate consequence of the definition is that the root of the data structure returned by an upwards accumulation is the fold of the original data structure, and it is a straightforward matter to calculate the more efficient characterization described in Theorem 8.

Definition 6. The function root_G :: L(A) → A returns the root of a homogeneous data structure:

root_G = fst ∘ out_G ⊓⊔ Lemma 7.

root_G ∘ ua_F ϕ = fold_F ϕ ⊓⊔ Proof. The proof depends on the universal property

h = unfold_G ϕ ⇔ out_G ∘ h = G(id, h) ∘ ϕ

of unfolds. We have

root_G ∘ ua_F ϕ = { root }

fst ∘ out_G ∘ ua_F ϕ = { universal property of unfold }

fst ∘ G(id, ua_F ϕ) ∘ (fold_F ϕ △ (F (one, id) ∘ out_F )) = { G(f, g) = f × F (id, g) }

fst ∘ (id × F (id, ua_F ϕ)) ∘ (fold_F ϕ △ (F (one, id) ∘ out_F )) = { pairs }

fold_F ϕ ⊓⊔ Theorem 8.

ua_F ϕ = fold_F (in_G ∘ (ϕ ∘ F (id, root)) △ F (one, id)) ⊓⊔ Proof. As well as properties of unfold, this proof depends also on the universal property

h = fold_F ϕ ⇔ h ∘ in_F = ϕ ∘ F (id, h)

of folds; in particular,

ua_F ϕ = fold_F ψ ⇔ ua_F ϕ ∘ in_F = ψ ∘ F (id, ua_F ϕ)

Using this property, we can calculate the appropriate ψ:

ua_F ϕ ∘ in_F = { isomorphism of datatypes: in_G ∘ out_G = id } in_G ∘ out_G ∘ ua_F ϕ ∘ in_F = { ua as an unfold }

in_G ∘ G(id, ua_F ϕ) ∘ (fold_F ϕ △ (F (one, id) ∘ out_F )) ∘ in_F = { G(f, g) = f × F (id, g); pairs }

in_G ∘ (fold_F ϕ △ (F (one, ua_F ϕ) ∘ out_F )) ∘ in_F = { composition distributes backwards over fork;

isomorphism of datatypes again } in_G ∘ ((fold_F ϕ ∘ in_F ) △ F (one, ua_F ϕ)) = { fold }

in_G ∘ ((ϕ ∘ F (id, fold_F ϕ)) △ F (one, ua_F ϕ)) = { Lemma 7 }

in_G ∘ ((ϕ ∘ F (id, root_G ∘ ua_F ϕ)) △ F (one, ua_F ϕ)) = { composition distributes backwards over fork } in_G ∘ ((ϕ ∘ F (id, root_G )) △ F (one, id)) ∘ F (id, ua_F ϕ)

Therefore

ua_F ϕ = fold_F ψ ⇐ ψ = in_G ∘ (ϕ ∘ F (id, root_G ) △ F (one, id)) ⊓⊔ With this improved characterization, the upwards accumulation ua_F ϕ takes asymptotically no longer to compute than the ordinary fold fold_F ϕ.

Example 9.

1. For cons lists, we have

uaL :: (Either () (a,b) -> b) -> List a -> Nlist b
uaL phi = foldL psi
  where
    psi (Left ()) = ConsN (phi (Left ())) (Left ())
    psi (Right (a,x))
                  = ConsN (phi (Right (a, rootN x))) (Right x)

Here, rootN returns the ‘root’ (the first element) of a non-empty list:

rootN :: Nlist a -> a
rootN (ConsN a x) = a

2. For leaf-labelled binary trees, we have

uaT :: (Either a (b,b) -> b) -> Ltree a -> Htree b
uaT phi = foldT psi
  where
    psi (Left a)
      = ConsH (phi (Left a)) (Left ())
    psi (Right (t,u))
      = ConsH (phi (Right (rootH t, rootH u))) (Right (t,u))

Here, rootH returns the root label of a homogeneous binary tree:

rootH :: Htree a -> a
rootH (ConsH a x) = a

3. For internally-labelled binary trees, we have

uaB :: (Either () (a,b,b) -> b) -> Btree a -> Htree b
uaB phi = foldB psi
  where
    psi (Left ())
      = ConsH (phi (Left ())) (Left ())
    psi (Right (a,t,u))
      = ConsH (phi (Right (a, rootH t, rootH u))) (Right (t,u))

4. For rose trees we have

uaR :: ((a, List b) -> b) -> Rtree a -> Rtree b
uaR phi = foldR psi
  where
    psi (a,us) = ConsR (phi (a, map rootR us)) us

Here, rootR returns the root label of a rose tree:

rootR :: Rtree a -> a
rootR (ConsR a ts) = a

⊓⊔ 5 Generic Downwards Accumulations

Just as with upwards accumulations, a downwards accumulation on a data structure returns a data structure of a possibly different type, namely the labelled variant of the original datatype. A downwards accumulation differs from an upwards accumulation in that it takes an extra accumulating parameter [2] with which to carry contextual information about its ancestors into a subtree. In this section, we will generalize the function daTree from Section 2 to a generic downwards accumulation for an arbitrary regular datatype.

Throughout this section, we assume that F is a bifunctor, T is its induced datatype, and L is the labelled variant (induced by bifunctor G).

5.1 The Essential Ideas

We will define a function da_F ϕ of type C × T (A) → L(B). Here, ϕ is the operation characterizing the particular accumulation, and C is the type of the accumulating parameter; the result is a labelled data structure with labels of type B.

We will write da_F ϕ as an unfold, unfold_G ψ for some ψ dependent on ϕ. Type considerations reveal that ψ therefore has type

ψ :: C × T (A) → G(B, C × T (A))

The only question now is how to define ψ. We will construct separately from a value of type C × T (A) values of type G(B, C ) and G(1, T (A)), and combine these two values into a single value of type G(B, C × T (A)).

In order to combine the two values, we suppose a function

zip_G :: G(A, B) × G(C , D) → G(A × C , B × D)

This function will be partial, defined only when both arguments have the same ‘shape’. In particular, when G constructs a sum type, these two arguments should be in the same choices of the sum. Zipping together G(B, C ) and G(1, T (A)) gives G(B×1, C ×T (A)); discarding the extra 1 with G(fst, id) yields the required G(B, C × T (A)), the result of ψ.

The second value is the easier to construct. Applying out_F to the second component of the input of type C ×T (A) yields a value of type C ×F (A, T (A)). Discarding the C and the A by applying one × F (one, id) produces the required type 1 × F (1, T (A)), or equivalently G(1, T (A)).

For the first value, we keep the root labels and discard the subtrees from the second component of the argument of type C ×T (A), by applying out_F and then F (id, one) to the second component. This produces values of type C ×F (A, T (A)) and then C × F (A, 1). Now we suppose that the argument ϕ to the downwards accumulation can complete the job:

ϕ :: C × F (A, 1) → B × F (1, C )

:: C × F (A, 1) → G(B, C )

Thus ϕ takes a pair as an argument and returns a pair as a result. The first component of the argument, of type C , is the accumulating parameter. The second component of the argument, of type F (A, 1), is the data structure with the recursive components stripped away, leaving only whatever labels are attached to this node. The first component of the result, of type B, is the label generated for this node of the result. The second component, of type F (1, C ), contains the accumulating parameters for the recursive calls, one for each subcomponent. There is an extra requirement, that the second component of the result have the same ‘shape’ as the second component of the argument, in order that the subsequent zip can combine the two.

Figure 2 illustrates the process. The dotted arrow is the product morphism induced by the two independent parts of the diagram.

5.2 Shape Preservation

The function ϕ used in a downwards accumulation has to be well-behaved with respect to shape, in a sense that we make precise here. Intuitively, ϕ must produce an appropriate collection of new accumulating parameters for the recursive calls, one accumulating parameter per call. In particular, when F yields a sum type, the accumulating parameters produced, of type F (1, C ), must be in the same variant of F as the argument, of type F (A, 1), consumed.

C × T (A)

C × F (A, 1) id × F (id, one) C × F (A, T (A)) id × outF

B × F (1, C ) = ϕ 1 × F (1, T (A)) = one × F (one, id)

G(B, C ) fst G(B, C ) × G(1, T (A)) ϕ ∘ (id × F (id, one)) △ one × F (one, id)

. . . . . . . . . . snd G(1, T (A))

G(B × 1, C × T (A)) zip_G

G(B, C × T (A))

G(fst, id)

Fig. 2. The anatomy of a downwards accumulation

We formalize this as follows. The shape of a data structure is obtained by discarding all the data in that data structure, retaining only structure. In particular, the shape of a value x of type F (A, B) is F (one, one) x, of type F (1, 1).

Definition 10.

shape_F = F (one, one) ⊓⊔ We require that if ϕ (c, x) = (b, y) then shape_F x = shape_F y.

5.3 Zipping Two Data Structures

One component of the construction outlined above is a function

zip_G :: G(A, B) × G(C , D) → G(A × C , B × D)

This function should be purely structural, not relying on the values of type A, B, C and D. In other words, it should be polymorphic, or a natural transformation [24], satisfying the free theorem

G(f × g, h × k) ∘ zip_G = zip_G ∘ (G(f, h) × G(g, k))

A function in the opposite direction is easy to define:

unzip_G :: G(A × C , B × D) → G(A, B) × G(C , D) unzip_G = G(fst, fst) △ G(snd, snd)

Another requirement of zip is that it should be a post-inverse of unzip:

zip_G ∘ unzip_G = id

The composition the other way round will in general not be the identity, because zip is usually a partial function: only data structures of the same shape can be zipped together. Note that unzip returns a pair of data structures of the same shape.

Hoogendijk [13] has made a thorough study of functions like zip (and generalizations involving functors other than ‘pair’) in the setting of relations. We merely assume the existence of a (possibly partial) natural transformation zip_G of the required type that is a post-inverse of unzip_G as defined above.

5.4 Putting it Together

To conclude, we have the following.

Definition 11. The function ϕ :: C ×F (A, 1)→B ×F (1, C ) is shape-preserving if, when ϕ (c, x) = (b, y), the shapes of x and y are the same: shape_F x = shape_F y; equivalently, if shape_F ∘ snd ∘ ϕ = shape_F ∘ snd. ⊓⊔

Definition 12. For shape-preserving ϕ of type C × F (A, 1) → B × F (1, C ), the downwards accumulation da_F ϕ :: C × T (A) → L(B) is defined by

da_F ϕ = unfold_G ψ

where

ψ :: C × T (A) → G(B, C × T (A)) ψ (c, t) = let u = out_F t v = ϕ (c, F (id, one) u) w = ((), F (one, id) u) in G(fst, id) (zip_G (v, w))

(The point-free characterization

ψ = G(fst, id) ∘ zip_G ∘ ((ϕ ∘ π_1) △ π_2) ∘ (id × out_F )

where

π_1 = id × F (id, one) π_2 = one × F (one, id)

is harder to comprehend, but easier to calculate with.) It can be computed in linear time, if ϕ takes constant time. ⊓⊔

Example 13. We can define a generic ‘depths’ function by

depths_F x = da_F ϕ (0, x)

where

ϕ (d, z) = (d + 1, F (one, const (d + 1)) z)

1. For cons lists we have

daL :: ((c, Either () a) -> (b, Either () c)) ->
       (c, List a) -> Nlist b
daL phi = unfoldN psi
  where
    psi :: (c, List a) -> (b, Either () (c, List a))
    psi (c, Nil) = (b, Left ())
                        where (b, Left ()) = phi (c, Left ())
    psi (c, Cons a x) = (b, Right (cx,x))
                        where (b, Right cx) = phi (c, Right a)

For example, the ‘depths’ function for lists is

depthsL :: List a -> Nlist Int
depthsL x = daL phi (0,x)
  where
    phi :: (Int, Either () a) -> (Int, Either () Int)
    phi (d, Left ()) = (d+1, Left ())
    phi (d, Right a) = (d+1, Right (d+1))

returning a list of the form [1,...,n+1] where the original list had n elements.

2. For non-empty lists we have

daN :: ((c, (a, Either () ())) -> (b, Either () c)) ->
       (c, Nlist a) -> Nlist b
daN phi = unfoldN psi
  where
    psi :: (c, Nlist a) -> (b, Either () (c, Nlist a))
    psi (c, ConsN a (Left ()))
      = (b, Left ())
        where (b,Left()) = phi (c, (a, Left ()))
    psi (c, ConsN a (Right x))
      = (b, Right (cx,x))
        where (b,Right cx) = phi (c, (a, Right()))

(Actually, strictly speaking, the shape-preservation requirement forces an extra unit into the result type of phi:

daN :: ((c, (a, Either () ())) -> (b, ((), Either () c))) ->
       (c, Nlist a) -> Nlist b

which would simply be discarded later:

... where (b,((),Left())) = phi (c, (a, Left ()))
... where (b,((),Right cx)) = phi (c, (a, Right()))

The definitions in this example are equivalent, but are closer to what a rational programmer would have written.) The depths function for non-empty lists is

depthsN :: Nlist a -> Nlist Int
depthsN x = daN phi (0,x)
  where
    phi :: (Int, (a, Either () ())) -> (Int, Either () Int)
    phi (d, (a, Left ())) = (d+1, Left ())
    phi (d, (a, Right ())) = (d+1, Right (d+1))

returning a list of the form [1,...,n] where the original list had n elements. 3. For leaf-labelled binary trees we have

daT :: ((c, Either a ()) -> (b, Either () (c,c))) ->
       (c, Ltree a) -> Htree b
daT phi = unfoldH psi
  where
    psi :: (c, Ltree a) -> (b, Either () ((c, Ltree a), (c, Ltree a)))
    psi (c,LeafT a) = (b, Left ())
                       where (b,Left ()) = phi (c, Left a)
    psi (c,BinT t u) = (b, Right ((ct,t),(cu,u)))
                       where (b,Right (ct,cu)) = phi (c, Right ())

The depths function for these trees is

depthsT :: Ltree a -> Htree Int
depthsT t = daT phi (0,t)
  where
    phi :: (Int, Either a ()) -> (Int, Either () (Int,Int))
    phi (d, Left a) = (d+1, Left ())
    phi (d, Right ()) = (d+1, Right (d+1,d+1))

4. For internally-labelled binary trees we have

daB :: ((c, Either () a) -> (b, Either () (c,c))) ->
       (c, Btree a) -> Htree b
daB phi = unfoldH psi
  where
    psi :: (c, Btree a) -> (b, Either () ((c, Btree a), (c, Btree a)))
    psi (c, Empty)
      = (b, Left ())
        where (b, Left ()) = phi (c, Left ())
    psi (c, BinB a t u)
      = (b, Right ((ct,t),(cu,u)))
        where (b, Right (ct,cu)) = phi (c, Right a)

The depths function for these trees is depthsB :: Btree a -> Htree Int
depthsB t = daB phi (0,t)
  where
    phi :: (Int, Either () a) -> (Int, Either () (Int,Int))
    phi (d, Left ()) = (d+1, Left ())
    phi (d, Right a) = (d+1, Right (d+1,d+1))

5. For homogeneous binary trees we have

daH :: ((c, (a, Either () ())) -> (b, Either () (c,c))) ->
       (c, Htree a) -> Htree b
daH phi = unfoldH psi
  where
    psi :: (c, Htree a) -> (b, Either () ((c, Htree a), (c, Htree a)))
    psi (c, ConsH a (Left ()))
      = (b, Left ())
        where (b, Left ()) = phi (c, (a, Left ()))
    psi (c, ConsH a (Right (t,u)))
      = (b, Right ((ct,t),(cu,u)))
        where (b, Right (ct,cu)) = phi (c, (a, Right ()))

The depths function for these trees is

depthsH :: Htree a -> Htree Int
depthsH t = daH phi (0,t)
  where
    phi :: (Int, (a, Either () ())) -> (Int, Either () (Int,Int))
    phi (d, (a, Left ())) = (d+1, Left ())
    phi (d, (a, Right ())) = (d+1, Right (d+1,d+1))

6. For rose trees we have

daR :: ((c,(a, List ())) -> (b, List c)) -> (c, Rtree a) -> Rtree b
daR phi = unfoldR psi
  where
    psi :: (c, Rtree a) -> (b, List (c, Rtree a))
    psi (c, ConsR a ts)
      = (b, zip cs ts)
        where (b,cs) = phi (c,(a, map (const ()) ts))

where zip :: List a -> List b -> List (a,b) zips together two lists of the same length to produce a list of pairs. The depths function for rose trees is

depthsR :: Rtree a -> Rtree Int
depthsR t = daR phi (0,t)
  where
    phi :: (Int, (a, List ())) -> (Int, List Int)
    phi (d, (a, zs)) = (d+1, map (const (d+1)) zs)

⊓⊔

Example 14. For homogeneous types, we can define a ‘paths’ function, labelling every node with the list of its ancestors. (For a heterogeneous type, we would need to label each node with a ‘heterogeneous list’, as different nodes can have different kinds of label. We might also want to record the ‘direction’ taken at each intermediate node to reach a particular node of the data structure. A characterization of paths incorporating these two refinements was used as the basis for defining downwards accumulations in [10], and is what made that approach more complicated than the one taken here.)

For homogeneous types, there is an isomorphism between F (A, B) and G(A, B). We write fg :: F (A, B) → G(A, B) for the isomorphism in one direction. Then we can define

paths_F x = da_F p ([ ], x)

where

p :: List(A) × F (A, 1) → List(A) × F (1, List(A)) p (x, z) = (x ++ [a], F (one, const (x ++ [a])) z) where a = fst (fg x)

1. For non-empty lists the paths function is

pathsN :: Nlist a -> Nlist [a]
pathsN x = daN phi ([],x)
  where
    phi :: ([a], (a, Either () ())) -> ([a], Either () [a])
    phi (p, (a, Left ())) = (p++[a], Left ())
    phi (p, (a, Right ())) = (p++[a], Right (p++[a]))

returning the initial segments of the list. 2. For homogeneous binary trees we have

pathsH :: Htree a -> Htree [a]
pathsH t = daH phi ([],t)
  where
    phi :: ([a], (a, Either () ())) -> ([a], Either () ([a],[a]))
    phi (p, (a, Left ())) = (p++[a], Left ())
    phi (p, (a, Right ())) = (p++[a], Right (p++[a],p++[a]))

3. For rose trees we have

pathsR :: Rtree a -> Rtree [a]
pathsR t = daR phi ([],t)
  where
    phi :: ([a], (a, List ())) -> ([a], List [a])
    phi (p, (a, zs)) = (p++[a], map (const (p++[a])) zs)

⊓⊔

6 Properties of Downwards Accumulations

In this section we collect a few theorems and properties of downwards accumulations. We present a universal property, a fusion law, two special cases of fusion with a map, and an accumulation lemma.

As with earlier sections, throughout this one we assume that F is a bifunctor, T is its induced datatype, and L is its labelled variant (induced by bifunctor G). Moreover, recall from Definition 12 that

da_F ϕ = unfold_G (G(fst, id) ∘ zip_G ∘ ((ϕ ∘ π_1) △ π_2) ∘ (id × out_F ))

where

π_1 = id × F (id, one) π_2 = one × F (one, id)

6.1 Universal Property

Recall the universal property of unfolds:

h = unfold_G ψ ⇔ out_G ∘ h = G(id, h) ∘ ψ

Because a downwards accumulation is simply an unfold, we obtain for free the following universal property for downwards accumulations.

Theorem 15.

h = da_F ϕ ⇔ out_G ∘ h = G(fst, h) ∘ zip_G ∘ ((ϕ ∘ π_1) △ π_2) ∘ (id × out_F ) ⊓⊔ Proof.

h = da_F ϕ ⇔ { da } h = unfold_G (G(fst, id) ∘ zip_G ∘ ((ϕ ∘ π_1) △ π_2) ∘ (id × out_F )) ⇔ { universal property of unfold }

out_G ∘ h = G(id, h) ∘ G(fst, id) ∘ zip_G ∘ ((ϕ ∘ π_1) △ π_2) ∘ (id × out_F ) ⇔ { functors }

out_G ∘ h = G(fst, h) ∘ zip_G ∘ ((ϕ ∘ π_1) △ π_2) ∘ (id × out_F ) ⊓⊔

6.2 Fusion Law

Likewise, corresponding to the fusion law for unfolds there is a fusion law for downwards accumulations. The fusion law for unfolds is as follows.

Lemma 16.

unfold_G ψ ∘ h = unfold_G ψ^′ ⇐ ψ ∘ h = G(id, h) ∘ ψ^′ ⊓⊔ Proof.

unfold_G ψ ∘ h = unfold_G ψ^′ ⇔ { universal property of unfolds } out_G ∘ unfold_G ψ ∘ h = G(id, unfold_G ψ ∘ h) ∘ ψ^′ ⇔ { unfold evaluation }

G(id, unfold_G ψ) ∘ ψ ∘ h = G(id, unfold_G ψ ∘ h) ∘ ψ^′ ⇐ { functors; Leibniz } ψ ∘ h = G(id, h) ∘ ψ^′ ⊓⊔ Instantiating both unfolds as downwards accumulations gives rise to the following fusion law for downwards accumulations.

Corollary 17.

da_F ϕ ∘ h = da_F ϕ^′ ⇐ zip_G ∘ ((ϕ ∘ π_1) △ π_2) ∘ (id × out_F ) ∘ h

= G(id, h) ∘ zip_G ∘ ((ϕ^′ ∘ π_1) △ π_2) ∘ (id × out_F ) ⊓⊔

6.3 Maps and Downwards Accumulations

Two common special cases of fusion are with a map, either before or after the downwards accumulation.

Theorem 18.

map_G f ∘ da_F ϕ = da_F (G(f, id) ∘ ϕ) ⊓⊔

Proof.

map_G f ∘ da_F ϕ = da_F ϕ^′ ⇔ { universal property }

out_G ∘ map_G f ∘ da_F ϕ

= G(fst, map_G f ∘ da_F ϕ) ∘ zip_G ∘ ((ϕ^′ ∘ π_1) △ π_2) ∘ (id × out_F ) ⇔ { evaluation for map and da }

G(f, map_G f) ∘ G(fst, da_F ϕ) ∘ zip_G ∘ ((ϕ ∘ π_1) △ π_2) ∘ (id × out_F ) = G(fst, map_G f ∘ da_F ϕ) ∘ zip_G ∘ ((ϕ^′ ∘ π_1) △ π_2) ∘ (id × out_F ) ⇐ { Leibniz }

G(f ∘ fst, id) ∘ zip_G ∘ (ϕ × id) = G(fst, id) ∘ zip_G ∘ (ϕ^′ × id) ⇐ { naturality of zip; pairs }

G(f, id) ∘ ϕ = ϕ^′ ⊓⊔ Theorem 19.

da_F ϕ ∘ (id × map_F f) = da_F (ϕ ∘ (id × F (f, id))) ⊓⊔ Proof.

da_F ϕ ∘ (id × map_F f) = da_F ϕ^′ ⇔ { universal property }

out_G ∘ da_F ϕ ∘ (id × map_F f)

= G(fst, da_F ϕ ∘ (id × map_F f)) ∘ zip_G ∘ ((ϕ^′ ∘ π_1) △ π_2) ∘ (id × out_F ) ⇔ { evaluation for da }

G(fst, da_F ϕ) ∘ zip_G ∘ ((ϕ ∘ π_1) △ π_2) ∘ (id × out_F ) ∘ (id × map_F f) = G(fst, da_F ϕ ∘ (id × map_F f)) ∘ zip_G ∘ ((ϕ^′ ∘ π_1) △ π_2) ∘ (id × out_F ) ⇐ { Leibniz; evaluation for map }

zip_G ∘ ((ϕ ∘ π_1) △ π_2) ∘ (id × (F (f, map_F f) ∘ out_F ))

= G(id, id × map_F f) ∘ zip_G ∘ ((ϕ^′ ∘ π_1) △ π_2) ∘ (id × out_F ) ⇐ { Leibniz; naturality of zip }

((ϕ ∘ π_1) △ π_2) ∘ (id × F (f, map_F f))

= (id × G(id, map_F f)) ∘ ((ϕ^′ ∘ π_1) △ π_2) ⇔ { pairs; relationship between F and G } ϕ ∘ π_1 ∘ (id × F (f, map_F f)) = ϕ^′ ∘ π_1 π_2 ∘ (id × F (f, map_F f)) = (id × F (id, map_F f)) ∘ π_2 ⇐ { π_1, π_2 }

ϕ ∘ (id × F (f, id)) = ϕ^′ (id × F (id, map_F f)) ∘ π_2 = (id × F (id, map_F f)) ∘ π_2 ⇐ { Leibniz }

ϕ ∘ (id × F (f, id)) = ϕ^′ ⊓⊔ 6.4 An Accumulation Lemma for Homogeneous Datatypes

We have seen in Example 14 that the ‘paths’ function on a homogeneous datatype can be expressed as a downwards accumulation. An accumulation lemma for downwards accumulations is a kind of converse of this observation: that (some) downwards accumulations can be expressed in terms of paths. Such lemmas have turned out to be very convenient for program calculation [4, 5, 1]. This section presents such a lemma.

Note that we have only a partial converse: not every downwards accumulation (even on homogeneous datatypes) can be expressed in terms of paths. In particular, the ‘paths’ function discards information distinguishing the different children of a node, so a downwards accumulation that treats a left child differently from a right child cannot be expressed in this way. The approach taken to defining downwards accumulations in [10], on the other hand, involved a polytypic definition of the ‘paths’ function that maintained all such information. As a consequence, in that paper the accumulation lemma applied to every downwards accumulation (indeed, that was how downwards accumulation were defined); however, the construction was much more complex than the one in this paper.

Definition 20. An accumulating fold over lists is defined as follows:

afold :: (B → A → B) → B → List(A) → B afold f e [ ] = e afold f e (x ++ [a]) = f (afold f e x) a ⊓⊔ The equation

afold f e ([a] ++ x) = afold f (f e a) x

is an alternative characterization of the behaviour on non-empty lists. This reveals the motivation for the name ‘accumulating fold’: the second argument acts as an accumulating parameter [2]. The same function is called foldl in Haskell.

Example 21. The length function on lists can be defined by

length = afold next 0 where next d a = d + 1 ⊓⊔ Theorem 22.

map_G (afold f e) (paths_F x) = da_F ϕ (e, x)

where

ϕ (u, z) = (f u a, F (one, const (f u a)) z) where a = fst (fg z) ⊓⊔ Example 23. The ‘depths’ function for a homogeneous datatype can be viewed as computing the lengths of every path:

depths_F = map_G length ∘ paths_F

Now Theorem 22 concludes that

depths_F x = da_F ϕ (0, x)

where

ϕ (d, z) = (d + 1, F (one, const (d + 1)) z)

—which is just the characterization of depths_F given in Example 13. ⊓⊔

Proof (of Theorem 22). Throughout this proof, we assume that p is as defined in Example 14. We have

map_G (afold f e) (paths_F x) = { paths }

map_G (afold f e) (da_F p ([ ], x)) = { fusion (Theorem 18) }

da_F (G(afold f e, id) ∘ p) ([ ], x)

Now we must extract the afold from the accumulation, so we explore the conditions under which we can do so.

da_F ϕ ∘ (afold f e × id) = da_F ϕ^′ ⇐ { fusion }

zip_G ∘ ((ϕ ∘ π_1) △ π_2) ∘ (id × out_F ) ∘ (afold f e × id) = G(id, afold f e × id) ∘ zip_G ∘ ((ϕ^′ ∘ π_1) △ π_2) ∘ (id × out_F ) ⇐ { naturality of zip; pairs; Leibniz }

((ϕ ∘ π_1) △ π_2) ∘ (afold f e × id)

= (G(id, afold f e) × id) ∘ ((ϕ^′ ∘ π_1) △ π_2) ⇔ { pairs again }

ϕ ∘ π_1 ∘ (afold f e × id) = G(id, afold f e) ∘ ϕ^′ ∘ π_1 π_2 ∘ (afold f e × id) = π_2 ⇐ { π_1 commutes with h × id; π_2 absorbs h × id }

ϕ ∘ (afold f e × id) = G(id, afold f e) ∘ ϕ^′

So in order to extract the afold from the accumulation, we must find a ϕ such that

ϕ ∘ (afold f e × id) = G(id, afold f e) ∘ G(afold f e, id) ∘ p

We start with the right-hand side:

(G(id, afold f e) ∘ G(afold f e, id) ∘ p) (x, z) = { let a = fst (fg z) }

G(afold f e, afold f e) (x ++ [a], F (one, const (x ++ [a])) z) = { relationship between G and F }

(afold f e (x ++ [a]), F (one, const (afold f e (x ++ [a]))) z) = { afold }

(f (afold f e x) a, F (one, const (f (afold f e x) a)) z) = { let ϕ (u, z) = (f u a, F (one, const (f u a)) z) where a = fst (fg z) } ϕ (afold f e x, z) obtaining the desired ϕ. Therefore,

map_G (afold f e) (paths_F x) = { first calculation above }

da_F (G(afold f e, id) ∘ p) ([ ], x) = { ϕ as above; extracting the afold } da_F ϕ (e, x) ⊓⊔ 7 Conclusion

We have shown how to generalize the notion of downwards accumulation [7, 8] to an arbitrary regular datatype, building on Bird et al.’s [1] generalization of upwards accumulation. The definition is as an unfold, further evidence of the usefulness [11] (and under-appreciation, not least on the author’s part, given that it has taken so long to discover this characterization) of this higher-order operator.

In an earlier version of this paper [10], we attempted to solve the problem of constructing a generic definition of downwards accumulation by considering the ‘paths’ to elements of an arbitrary data structure. The path to an element records the ancestors of that element in the data structure; a downwards accumulation consists of a fold mapped over the paths. (Symmetrically, the descendants of an element of a data structure form a ‘subtree’ of that data structure, and an upwards accumulation consists of a fold mapped over the subtrees of the data structure.) We do something similar but much simpler in Example 14: there we model a path merely as a list of elements, whereas in [10] we had to include also the ‘directions’ along a path and to consider nodes with different kinds of label.

In retrospect, it appears that starting with paths was the wrong way to approach the problem. Modelling the paths of an arbitrary datatype involves ‘linearizing’ the type functor for that type, among other reasons to determine the branching degree of a node. This was quite a complicated construction; the approach taken in this paper is much simpler.

Another advantage of the approach presented here is that it leads to a generic or parametric higher-order polymorphic instead of a polytypic or ad-hoc higherorder polymorphic construction. By the former we mean a construction based on semantic properties of the type functors concerned, such as Bird et al.’s labelling construction G(A, X ) = A × F (1, X ). By the latter we mean a construction by induction over the syntactic presentation of the type functor, in the style of Jeuring [14]. (In fact, our characterization in [10] was not even as general as Jeuring’s scheme: we required the functor to be polynomial, that is, a sum of products, rather than regular as Jeuring allows.) Hoogendijk [12, 13] argues for the inherent superiority of generic over polytypic definitions, but for us there is the added bonus of applicability to regular but non-polynomial datatypes such as rose trees.

Acknowledgements

The members of the Problem Solving Club at Oxford University Computing Laboratory, especially Richard Bird, Jesús Ravelo and Pedro Borges, have made many helpful suggestions and comments; Ross Paterson and the anonymous Mathematics of Program Construction and Science of Computer Programming referees have also suggested many improvements. The commuting diagrams were drawn using Paul Taylor’s L^ATEX macros.

References

1. Richard Bird, Oege de Moor, and Paul Hoogendijk. Generic functional programming with types and relations. Journal of Functional Programming, 6(1):1–28, 1996.

2. Richard S. Bird. The promotion and accumulation strategies in transformational programming. ACM Transactions on Programming Languages and Systems, 6(4):487–504, October 1984. See also [3].

3. Richard S. Bird. Addendum to “The promotion and accumulation strategies in transformational programming”. ACM Transactions on Programming Languages and Systems, 7(3):490–492, July 1985.

4. Richard S. Bird. An introduction to the theory of lists. In M. Broy, editor, Logic of Programming and Calculi of Discrete Design, pages 3–42. Springer-Verlag, 1987. Also available as Technical Monograph PRG-56, from the Programming Research Group, Oxford University.

5. Richard S. Bird, Jeremy Gibbons, and Geraint Jones. Formal derivation of a pattern matching algorithm. Science of Computer Programming, 12(2):93–104, July 1989.

6. Maarten M. Fokkinga and Erik Meijer. Program calculation properties of continuous algebras. Technical Report CS-R9104, CWI, Amsterdam, January 1991.

7. Jeremy Gibbons. Algebras for Tree Algorithms. D. Phil. thesis, Programming Research Group, Oxford University, 1991. Available as Technical Monograph PRG- 94. 8. Jeremy Gibbons. Upwards and downwards accumulations on trees. In R. S. Bird, C. C. Morgan, and J. C. P. Woodcock, editors, LNCS 669: Mathematics of Program Construction, pages 122–138. Springer-Verlag, 1993. A revised version appears in the Proceedings of the Massey Functional Programming Workshop, 1992.

9. Jeremy Gibbons. Deriving tidy drawings of trees. Journal of Functional Programming, 6(3):535–562, 1996.

10. Jeremy Gibbons. Polytypic downwards accumulations. In Johan Jeuring, editor, LNCS 1422: Proceedings of Mathematics of Program Construction, Marstrand, Sweden, June 1998. Springer-Verlag.

11. Jeremy Gibbons and Geraint Jones. The under-appreciated unfold. In Proceedings of the Third ACM SIGPLAN International Conference on Functional Programming, pages 273–279, Baltimore, Maryland, September 1998.

12. Paul Hoogendijk. A Generic Theory of Datatypes. PhD thesis, TU Eindhoven, 1997.

13. Paul Hoogendijk and Roland Backhouse. When do datatypes commute? In Eugenio Moggi and Guiseppe Rosolini, editors, LNCS 1290: Category Theory and Computer Science, pages 242–260. Springer-Verlag, September 1997.

14. Patrick Jansson and Johan Jeuring. PolyP – a polytypic programming language extension. In Conference Record of POPL ’97: The 24th ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages, pages 470–482, 1997.

15. Richard E. Ladner and Michael J. Fischer. Parallel prefix computation. Journal of the ACM, 27(4):831–838, October 1980.

16. Ian A. MacLeod. A query language for retrieving information from hierarchical text structures. Computer Journal, 34(3):254–264, 1991.

17. Grant Malcolm. Data structures and program transformation. Science of Computer Programming, 14:255–279, 1990.

18. Lambert Meertens. First steps towards the theory of rose trees. CWI, Amsterdam;

IFIP Working Group 2.1 working paper 592 ROM-25, 1988.

19. Lambert Meertens. Paramorphisms. Formal Aspects of Computing, 4(5):413–424, 1992.

20. John Peterson, Kevin Hammond, Lennart Augustsson, Brian Boutel, Warren Burton, Joseph Fasel, Andrew D. Gordon, John Hughes, Paul Hudak, Thomas Johnsson, Mark Jones, Erik Meijer, Simon Peyton Jones, Alastair Reid, and Philip Wadler. The Haskell 1.4 report. http://www.haskell.org/report/, April 1997.

21. Edward M. Reingold and John S. Tilford. Tidier drawings of trees. IEEE Transactions on Software Engineering, 7(2):223–228, March 1981.

22. John C. Reynolds. Semantics of the domain of flow diagrams. Journal of the ACM, 24(3):484–503, 1977.

23. David Skillicorn. Foundations of Parallel Programming. Number 6 in International Series on Parallel Computation. Cambridge University Press, 1994.

24. Philip Wadler. Theorems for free! In Functional Programming Languages and Computer Architecture, pages 347–359. ACM, 1989.